Double Pendulum

Equations of Motion

This notebook shows derivation of the equations of motion for the double pendulum simulation. See `http://www.myphysicslab.com/dbl_pendulum.html`.

Author: Erik Neumann

We regard y as increasing upwards. Origin is at the fixed base pivot.
$\theta 1$ is the angle at the (fixed) base pivot between the rod and the downward vertical position.
$\theta 2$ is the angle between the downward vertical from mass 1 and the second rod.
d is the derivative operator with respect to time.
L1, L2 are the lengths of the rods.
m1, m2 are masses.
g is gravitational constant.
x1,y1 is position of mass 1.
x2,y2 is position of mass2.
We start with the following:
x1 = L1 Sin[θ1]
y1 = −L1 Cos[θ1]
x2 = x1 + L2 Sin[θ2]
y2 = y1 − L2 Cos[θ2]
Then take two derivatives to get ddθ1 and ddθ2 as given below.

Naming convention: the letter d indicates derivative (with respect to time), dd indicates second derivative. For example:
x1' = dx1
x1'' = ddx1
θ1' = dθ1
θ1'' = ddθ1

```
ddx1 = -dθ1^2 L1 Sin[θ1] + ddθ1 L1 Cos[θ1]
```

```
General::spell1 :
 Possible spelling error: new symbol name "ddθ1" is similar to existing symbol "dθ1".
```

```
ddθ1 L1 Cos[θ1] - dθ1^2 L1 Sin[θ1]
```

```
ddy1 = dθ1^2 L1 Cos[θ1] + ddθ1 L1 Sin[θ1]
```

```
General::spell1 :
 Possible spelling error: new symbol name "ddy1" is similar to existing symbol "ddx1".
```

```
dθ1^2 L1 Cos[θ1] + ddθ1 L1 Sin[θ1]
```

```
ddx2 = ddx1 - dθ2^2 L2 Sin[θ2] + ddθ2 L2 Cos[θ2]
```

```
General::spell1 :
 Possible spelling error: new symbol name "ddθ2" is similar to existing symbol "dθ2".
```

```
ddθ1 L1 Cos[θ1] + ddθ2 L2 Cos[θ2] - dθ1^2 L1 Sin[θ1] - dθ2^2 L2 Sin[θ2]
```

```
ddy2 = ddy1 + dθ2^2 L2 Cos[θ2] + ddθ2 L2 Sin[θ2]
```

```
General::spell1 :
 Possible spelling error: new symbol name "ddy2" is similar to existing symbol "ddx2".
```

```
dθ1^2 L1 Cos[θ1] + dθ2^2 L2 Cos[θ2] + ddθ1 L1 Sin[θ1] + ddθ2 L2 Sin[θ2]
```

```
eq1 = -Cos[θ1] (m1 ddx1 + m2 ddx2) == Sin[θ1] (m1 ddy1 + m2 ddy2 + m2 g + m1 g)

-Cos[θ1] (m1 (ddθ1 L1 Cos[θ1] - dθ1^2 L1 Sin[θ1]) +
    m2 (ddθ1 L1 Cos[θ1] + ddθ2 L2 Cos[θ2] - dθ1^2 L1 Sin[θ1] - dθ2^2 L2 Sin[θ2])) ==
 Sin[θ1] (g m1 + g m2 + m1 (dθ1^2 L1 Cos[θ1] + ddθ1 L1 Sin[θ1]) +
    m2 (dθ1^2 L1 Cos[θ1] + dθ2^2 L2 Cos[θ2] + ddθ1 L1 Sin[θ1] + ddθ2 L2 Sin[θ2]))

eq2 = -Cos[θ2] m2 ddx2 == Sin[θ2] (m2 ddy2 + m2 g)

-m2 Cos[θ2] (ddθ1 L1 Cos[θ1] + ddθ2 L2 Cos[θ2] - dθ1^2 L1 Sin[θ1] - dθ2^2 L2 Sin[θ2]) ==
 Sin[θ2] (g m2 + m2 (dθ1^2 L1 Cos[θ1] + dθ2^2 L2 Cos[θ2] + ddθ1 L1 Sin[θ1] + ddθ2 L2 Sin[θ2]))

Solve[{eq1, eq2}, {ddθ1, ddθ2}] // Simplify

{{ddθ1 ->
   -(g (2 m1 + m2) Sin[θ1] + m2 (g Sin[θ1 - 2 θ2] + 2 (dθ2^2 L2 + dθ1^2 L1 Cos[θ1 - θ2]) Sin[θ1 - θ2])) /
    (L1 (2 m1 + m2 - m2 Cos[2 (θ1 - θ2)])),
  ddθ2 -> (2 (dθ1^2 L1 (m1 + m2) + g (m1 + m2) Cos[θ1] + dθ2^2 L2 m2 Cos[θ1 - θ2]) Sin[θ1 - θ2]) /
    (L2 (2 m1 + m2 - m2 Cos[2 (θ1 - θ2)]))}}
```